
1. [bookmark: _Toc122434485][bookmark: _Toc100053404]Overview
This document lists all the essential changes needed to correct problems in the TTCN implementation of test case 10.2 which is part of the NR5GC test suite. 
With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence. 
[bookmark: _Hlk90298041]Contact:	Kalahasti, Narendra
	Narendra.Kalahasti@anritsu.com
0. [bookmark: _Toc100053405]Verification Test Summary 
Test Case: 	10.2
ATS Version:	iwd-TTCN3-B2020-09_D22wk12
System Simulator used:	Anritsu Protocol Conformance Test System ME7834NR,
UE used:	Qualcomm SM 8450+SDX65 5G UE
Verification Status:	PASS

1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc100053406]Corrections required
1. [bookmark: _Toc30685521][bookmark: _Toc97137453][bookmark: _Toc100053407]Change 1
	Function name
	f_TC_10_1and2_NR5GC_IMS2

	Reason for change
	1. Test 10.2 is emergency call with registration therefore condition A7 is applied instead of A6 when calling f_IMS_EmergencyCallSetup_NR5GC_A_6_Step1 function.
2. It is necessary to coordinate resource reservation (PDU session Modification) with the call establishment, otherwise depending on the timing, it could happen that resources reservation starts when the call is already released.
3. An incoming coordination from IPCAN is not handled after call is released
4. f_NR5GC_EmergencyCallRelease is waiting for a trigger from IMS component, that is currently not sending it.

	Summary of change
	1. Changed condition A6 to A7 when calling f_IMS_EmergencyCallSetup_NR5GC_A_6_Step1.
2. Added ‘WaitForTrigger’ from IPCAN (trigger added into f_NR5GC_EmergencyCallPDUSession – see change 2) so the call can be released only after IPCAN is done with resource reservation.
3. f_IMS_CC_EndCall – added to handle trigger from IPCAN
4. Added f_IMS_IPCAN_SendCoOrdMsg(IPCAN) to be handled by f_NR5GC_EmergencyCallRelease

	TTCN module
	IMS\NR5GC_10\IMS_NR5GC_EmergencyCallTestcases.ttcn

	MCC160 Comment
	1. Accepted
2. Not needed.  See change 2 for alternative solution
3.Alternative solution implemented: only send the trigger from f_NR5GC_EmergencyCallRelease if required – and set this parameter to false when called from f_NR5GC_IPCAN_ProcedureEnd
4.Alternative solution implemented: Send the trigger from f_IMS_CC_Postamble if it’s an emergency call in NR5GC



Before Change
	  function f_TC_10_1and2_NR5GC_IMS2(boolean p_GeolocationUsage := false) runs on IMS_PTC
  { /* IMS implementation for PDN2 - Emergency call */
    var SipUrl v_ContactUrlUE;
    var IMS_DATA_REQ v_IMS_DATA_REQ;
    var REGISTER_Request v_RegisterRequest;
    var boolean v_CondA7 := true;
    var boolean v_GeolocationUsage := p_GeolocationUsage; /* the test case intends to use GeoLocationInformation if supported by the UE (PICS)
                                                             INVITE (step 1 of Annex C.22) refers to condition A8 of A.2.1 */
    f_IMS_PTC_Init(PDN_2);
    
    // Test Case Body
    f_IMS_TestBody_Set(true);
    
    f_IMS_IMS_WaitForTrigger(OtherPDN);
    // Steps 1-8 of Table 4.9.11.2.2-1 of TS 38.508-1 are performed.
    // EXCEPTION: In parallel to the events described in steps 9-10 below the events specified in step of Table 4.9.11.2.2-2 of TS 38.508-1 [21] take place
    // Steps 9-10 of Table 4.9.11.2.2-1 of TS 38.508-1 are performed.
    //EXCEPTION: In parallel to the events described in steps 12-20  below the events specified in 11-13 of annex Table 4.9.11.2.2-1 of TS 38.508-1[21] are performed.
    f_IMS_CC_StartCall(IPCAN_EmergencyCall_NoRegistration); // 4.9.11.2.2-2 of TS 38.508-1
    
    //@siclog "Step 2" siclog@
    //Step 1 of annex A.3 (emergency registration)
    //Check: Does the UE send a correctly composed initial REGISTER request for IMS emergency registration?
    v_IMS_DATA_REQ := f_IMS_REGISTER_InitialRequest(-, omit, -, -, -, v_CondA7);
    v_RegisterRequest := v_IMS_DATA_REQ.Request.Register;
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 2");
    
    // Reconfig of UE client ports in case the UE uses UDP and different than default (5060) ports
    f_IMS_Register_SecurityInit(v_RegisterRequest);
    
    //@siclog "Step 3" siclog@
    //Step 2 of annex A.3 (emergency registration)
    IMS_Server.send(cas_IMS_DATA_RSP(f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo), cs_Response(c_statusLine401, f_IMS_RegisterResponse_401_MessageHeaderTX(v_RegisterRequest))));

    //@siclog "Step 4" siclog@
    //Step 3 of annex A.3 (emergency registration)
    //Check: Does the UE correctly authenticate itself by sending another REGISTER request with a correctly composed Authorization header using the AKAv1-MD5 algorithm?
    v_IMS_DATA_REQ := f_IMS_REGISTER_SubsequentRequest(-, -, v_CondA7);
    v_RegisterRequest := v_IMS_DATA_REQ.Request.Register;
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 4");
    
    //@siclog "Step 5" siclog@
    //Step 4 of annex A.3 (emergency registration)
    v_RegisterRequest := v_IMS_DATA_REQ.Request.Register;
    IMS_Server.send(cas_IMS_DATA_RSP(f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo), cs_Response(c_statusLine200, f_IMS_RegisterResponse_200_MessageHeaderTX(v_RegisterRequest, true))));
  
    //@siclog "Step 6" siclog@
    //Step 1 of annex A.6 (emergency call)
    //Check: Does the UE send a correctly composed INVITE request?
    v_IMS_DATA_REQ := f_IMS_EmergencyCallSetup_NR5GC_A_6_Step1(A_2_1_A6,  v_GeolocationUsage);
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 6");

    //@siclog "Step 7" siclog@
    //Step 2 of annex A.6 (emergency call)
    //@siclog "Step 8" siclog@
    //Step 3 of annex A.6 (emergency call)
    //@siclog "Step 9" siclog@
    //Step 4 of annex A.6 (emergency call)
    //@siclog "Step 10" siclog@
    //Step 5 of annex A.6 (emergency call)
    //Check: Does the UE send ACK?
    f_IMS_EmergencyCallSetup_NR5GC_A_6_Step2ToEnd(v_IMS_DATA_REQ,  v_GeolocationUsage);
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 10");
    


    //@siclog "Step 11" siclog@
    //Step 1 of annex A.8 (MT Release of Voice Call)
    //@siclog "Step 12" siclog@
    //Step 2 of annex A.8 (MT Release of Voice Call)
    //Check: Does the UE send 200 OK for the BYE request and ends the call?
    v_ContactUrlUE := f_IMS_PTC_ImsInfo_GetContactUrl(dialog);
    f_IMS_CallReleaseMT(v_ContactUrlUE);

    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 12");
    
    f_IMS_TestBody_Set(false);
    



    f_IMS_IMS_SendCoOrdMsg(OtherPDN);
  }



After Change
	    function f_TC_10_1and2_NR5GC_IMS2(boolean p_GeolocationUsage := false) runs on IMS_PTC
  { /* IMS implementation for PDN2 - Emergency call */
    var SipUrl v_ContactUrlUE;
    var IMS_DATA_REQ v_IMS_DATA_REQ;
    var REGISTER_Request v_RegisterRequest;
    var boolean v_CondA7 := true;
    var boolean v_GeolocationUsage := p_GeolocationUsage; /* the test case intends to use GeoLocationInformation if supported by the UE (PICS)
                                                             INVITE (step 1 of Annex C.22) refers to condition A8 of A.2.1 */
    f_IMS_PTC_Init(PDN_2);
    
    // Test Case Body
    f_IMS_TestBody_Set(true);
    
    f_IMS_IMS_WaitForTrigger(OtherPDN);
    // Steps 1-8 of Table 4.9.11.2.2-1 of TS 38.508-1 are performed.
    // EXCEPTION: In parallel to the events described in steps 9-10 below the events specified in step of Table 4.9.11.2.2-2 of TS 38.508-1 [21] take place
    // Steps 9-10 of Table 4.9.11.2.2-1 of TS 38.508-1 are performed.
    //EXCEPTION: In parallel to the events described in steps 12-20  below the events specified in 11-13 of annex Table 4.9.11.2.2-1 of TS 38.508-1[21] are performed.
    f_IMS_CC_StartCall(IPCAN_EmergencyCall_NoRegistration); // 4.9.11.2.2-2 of TS 38.508-1
    
    //@siclog "Step 2" siclog@
    //Step 1 of annex A.3 (emergency registration)
    //Check: Does the UE send a correctly composed initial REGISTER request for IMS emergency registration?
    v_IMS_DATA_REQ := f_IMS_REGISTER_InitialRequest(-, omit, -, -, -, v_CondA7);
    v_RegisterRequest := v_IMS_DATA_REQ.Request.Register;
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 2");
    
    // Reconfig of UE client ports in case the UE uses UDP and different than default (5060) ports
    f_IMS_Register_SecurityInit(v_RegisterRequest);
    
    //@siclog "Step 3" siclog@
    //Step 2 of annex A.3 (emergency registration)
    IMS_Server.send(cas_IMS_DATA_RSP(f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo), cs_Response(c_statusLine401, f_IMS_RegisterResponse_401_MessageHeaderTX(v_RegisterRequest))));

    //@siclog "Step 4" siclog@
    //Step 3 of annex A.3 (emergency registration)
    //Check: Does the UE correctly authenticate itself by sending another REGISTER request with a correctly composed Authorization header using the AKAv1-MD5 algorithm?
    v_IMS_DATA_REQ := f_IMS_REGISTER_SubsequentRequest(-, -, v_CondA7);
    v_RegisterRequest := v_IMS_DATA_REQ.Request.Register;
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 4");
    
    //@siclog "Step 5" siclog@
    //Step 4 of annex A.3 (emergency registration)
    v_RegisterRequest := v_IMS_DATA_REQ.Request.Register;
    IMS_Server.send(cas_IMS_DATA_RSP(f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo), cs_Response(c_statusLine200, f_IMS_RegisterResponse_200_MessageHeaderTX(v_RegisterRequest, true))));
  
    //@siclog "Step 6" siclog@
    //Step 1 of annex A.6 (emergency call)
    //Check: Does the UE send a correctly composed INVITE request?
    v_IMS_DATA_REQ := f_IMS_EmergencyCallSetup_NR5GC_A_6_Step1(A_2_1_A7,  v_GeolocationUsage); //A7 = with registration (A6 = without)
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 6");

    //@siclog "Step 7" siclog@
    //Step 2 of annex A.6 (emergency call)
    //@siclog "Step 8" siclog@
    //Step 3 of annex A.6 (emergency call)
    //@siclog "Step 9" siclog@
    //Step 4 of annex A.6 (emergency call)
    //@siclog "Step 10" siclog@
    //Step 5 of annex A.6 (emergency call)
    //Check: Does the UE send ACK?
    f_IMS_EmergencyCallSetup_NR5GC_A_6_Step2ToEnd(v_IMS_DATA_REQ,  v_GeolocationUsage);
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 10");
    
    f_IMS_IPCAN_WaitForTrigger(IPCAN);

    //@siclog "Step 11" siclog@
    //Step 1 of annex A.8 (MT Release of Voice Call)
    //@siclog "Step 12" siclog@
    //Step 2 of annex A.8 (MT Release of Voice Call)
    //Check: Does the UE send 200 OK for the BYE request and ends the call?
    v_ContactUrlUE := f_IMS_PTC_ImsInfo_GetContactUrl(dialog);
    f_IMS_CallReleaseMT(v_ContactUrlUE);
    f_IMS_CC_EndCall(IPCAN_EmergencyCall_NormalService, IPCAN_SpeechAndEmergencyCall); //To receive coordination NR5GC
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 12");
    
    f_IMS_TestBody_Set(false);
    
    // Inform NR that call is released (caught by f_NR5GC_EmergencyCallRelease)
    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);

    f_IMS_IMS_SendCoOrdMsg(OtherPDN);
  }



1. [bookmark: _Toc100053408]Change 2
	Function name
	f_NR5GC_EmergencyCallPDUSession

	Reason for change
	This is coordination counter-part for Change 1 bullet 2.

	Summary of change
	Added coordination trigger which is awaited by IMS component as per Change 1 bullet 2.

	TTCN module
	NR5GC_IMSProcedures.ttcn

	MCC160 Comment
	In order not to impact 38.523-1 TCs which have MO call release, an alternative solution implemented: 
Added the sending of a trigger in f_NR5GC_IPCAN_EmergencyCall and waiting for the trigger in f_IMS_EmergencyCallSetup_NR5GC_A_6_Step2ToEnd



Before Change
	function f_NR5GC_EmergencyCallPDUSession(NR_CellId_Type p_NR_CellId,
                                           NR_RadioBearerId_Type p_SRB := tsc_NR_RbId_SRB2,
                                           integer p_PDUEstAlreadyRxd := 0) runs on NR5GC_PTC
  {
    var template (omit) GSM_MobilityInfo_Type v_GSM_MobilityInfo;

    f_NR5GC_PDUSessionEstablishment(p_NR_CellId,
                                    1,
                                    p_SRB, -,                       // @sic R5s210518 sic@
                                    omit,
                                    omit,
                                    crs_SSC_Mode(-, '001'B),        // @sic R5s210518 sic@
                                    -,
                                    f_BuildDefaultQoSRules("2"),
                                    -, -, -, -, -, -,
                                    crs_SSC_Mode('A'H, '001'B),        // @sic R5s210518, R5s210771 sic@
                                    -, -, -, -, -, -, -,
                                    cr_NG_Request_Type('011'B),
                                    -, -, -, -, -, -, -, -,
                                    "2",
                                    -,
                                    noRrcConnectionRelease,        // @sic R5s210515 sic@ don't need the extra delay here as there's further IMS signalling
                                    p_PDUEstAlreadyRxd);           // @sic R5-217286 sic@

    // Get the Info for the PDU Session just created
    v_GSM_MobilityInfo := f_Get_PDUSessionForDNNType(f_NR5GC_MobileInfo_GetSessionInfoList(), Emergency_PDN);
    
    // Wait until need to add voice QoS
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS2]);
    f_NR5GC_ModifyPDUSession_AddVoice(p_NR_CellId, valueof(v_GSM_MobilityInfo));


    // @sic R5s210518 sic@
  }



After Change
	    function f_NR5GC_EmergencyCallPDUSession(NR_CellId_Type p_NR_CellId,
                                           NR_RadioBearerId_Type p_SRB := tsc_NR_RbId_SRB2,
                                           integer p_PDUEstAlreadyRxd := 0) runs on NR5GC_PTC
  {
    var template (omit) GSM_MobilityInfo_Type v_GSM_MobilityInfo;

    f_NR5GC_PDUSessionEstablishment(p_NR_CellId,
                                    1,
                                    p_SRB, -,                       // @sic R5s210518 sic@
                                    omit,
                                    omit,
                                    crs_SSC_Mode(-, '001'B),        // @sic R5s210518 sic@
                                    -,
                                    f_BuildDefaultQoSRules("2"),
                                    -, -, -, -, -, -,
                                    crs_SSC_Mode('A'H, '001'B),        // @sic R5s210518, R5s210771 sic@
                                    -, -, -, -, -, -, -,
                                    cr_NG_Request_Type('011'B),
                                    -, -, -, -, -, -, -, -,
                                    "2",
                                    -,
                                    noRrcConnectionRelease,        // @sic R5s210515 sic@ don't need the extra delay here as there's further IMS signalling
                                    p_PDUEstAlreadyRxd);           // @sic R5-217286 sic@

    // Get the Info for the PDU Session just created
    v_GSM_MobilityInfo := f_Get_PDUSessionForDNNType(f_NR5GC_MobileInfo_GetSessionInfoList(), Emergency_PDN);
    
    // Wait until need to add voice QoS
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS2]);
    f_NR5GC_ModifyPDUSession_AddVoice(p_NR_CellId, valueof(v_GSM_MobilityInfo));
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS2]);

    // @sic R5s210518 sic@
  }



1. [bookmark: _Toc100053409]Change 3
	Function name
	f_TC_10_2_NR5GC_IMS1

	Reason for change
	The f_IMS_CC_Postamble (if called with TestProcedure parameter) is informing IPCAN to release PDU session, however that was already done from IMS2 and doing it again from IMS1 would lead to runtime error as there are no more PDU sessions.

	Summary of change
	f_IMS_CC_Postamble is not called with the TestProcedure parameter.

	TTCN module
	IMS_NR5GC_EmergencyCallTestcases.ttcn

	MCC160 Comment
	Alternative solution implemented: moved existing function call from IMS1 to IMS2



Before Change
	  function f_TC_10_2_NR5GC_IMS1() runs on IMS_PTC
  { // Emergency Call with emergency registration / Success / Location information not available / 5GS
    
    f_IMS_CC_Preamble(IPCAN_EmergencyCall, IMS_REGISTERED);
          
    f_IMS_IMS_SendCoOrdMsg(OtherPDN);
    
    f_IMS_IMS_WaitForTrigger(OtherPDN); // wait for trigger after IMS call release
    
    f_IMS_CC_Postamble(IPCAN_EmergencyCall_NormalService);
  }



After Change
	    function f_TC_10_2_NR5GC_IMS1() runs on IMS_PTC
  { // Emergency Call with emergency registration / Success / Location information not available / 5GS
    
    f_IMS_CC_Preamble(IPCAN_EmergencyCall, IMS_REGISTERED);
          
    f_IMS_IMS_SendCoOrdMsg(OtherPDN);
    
    f_IMS_IMS_WaitForTrigger(OtherPDN); // wait for trigger after IMS call release
    
    f_IMS_CC_Postamble(); 
  }



1. [bookmark: _Toc100053410]Change 4
	Testcase name
	TC_10_2_NR5GC

	Reason for change
	Adding and starting of IMS2 component is missing in testcase definition.

	Summary of change
	Added creation and start of IMS2.

	TTCN module
	IMS_Testsuite_NR5GC.ttcn

	MCC160 Comment
	Accepted



Before Change
	testcase TC_10_2_NR5GC() runs on MTC_IMS system SYSTEM_IMS {
    // @purpose
    //   Emergency Call with emergency registration / Success / Location information not available / 5GS
    var IMS_PTC          v_IMS1       := null;
    var IMS_PTC          v_IMS2       := null;
  
    timer t_GuardTimer := int2float(300);
  
    v_IMS1 := IMS_PTC.create("IMS1") alive;
  
    f_MTC_IMS_CreateMapAndConnectPTCs_NR5GC(system, v_IMS1, v_IMS2);
  
    v_IMS1.start(f_TC_10_2_NR5GC_IMS1());
  
    t_GuardTimer.start;
  
    f_MTC_MainLoop_IMS(t_GuardTimer);
  }



After Change
	    testcase TC_10_2_NR5GC() runs on MTC_IMS system SYSTEM_IMS {
    // @purpose
    //   Emergency Call with emergency registration / Success / Location information not available / 5GS
    var IMS_PTC          v_IMS1       := null;
    var IMS_PTC          v_IMS2       := null;
  
    timer t_GuardTimer := int2float(300);
  
    v_IMS1 := IMS_PTC.create("IMS1") alive;
    v_IMS2 := IMS_PTC.create("IMS2") alive;
  
    f_MTC_IMS_CreateMapAndConnectPTCs_NR5GC(system, v_IMS1, v_IMS2);
  
    v_IMS1.start(f_TC_10_2_NR5GC_IMS1());
    v_IMS2.start(f_TC_10_2_NR5GC_IMS2());
  
    t_GuardTimer.start;
  
    f_MTC_MainLoop_IMS(t_GuardTimer);
  }



MCC160 Impementation:
  function f_NR5GC_IPCAN_EmergencyCall(IMS_IPCAN_CO_ORD_PORT p_Port,
                                       NR_CellId_Type p_NR_CellId,
                                       boolean p_ConnectionEstablished) runs on NR5GC_PTC
  {
    var integer v_PDUEstAlreadyRxd := 0;
    if (p_ConnectionEstablished) {
      f_NR5GC_EmergencyCallPDUSession(p_NR_CellId);
    } else { // @sic R5-217857 sic@
      v_PDUEstAlreadyRxd := f_NR5GC_EmergencyCallWithRegistration_Steps1_8 (p_NR_CellId);
      f_NR5GC_EmergencyCallPDUSession(p_NR_CellId, -, v_PDUEstAlreadyRxd);
    }
    f_IMS_IPCAN_SendCoOrdMsg(p_Port); 
  }

  function f_IMS_EmergencyCallSetup_NR5GC_A_6_Step2ToEnd(IMS_DATA_REQ p_IMS_DATA_REQ,
                                    boolean p_GeolocationInfoAvailable := false) runs on IMS_PTC return INVITE_Request  // @sic R5-220214 sic@
  {
    var INVITE_Request v_INVITE_Request;
    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);
    v_INVITE_Request := return f_IMS_EmergencyCallSetup_Step2ToEnd (p_IMS_DATA_REQ, p_GeolocationInfoAvailable);
    f_IMS_IPCAN_WaitForTrigger(IPCAN);  
    return v_INVITE_Request;

  }

  function f_NR5GC_EmergencyCallRelease(NR_CellId_Type p_NR_CellId , boolean p_MOTriggerRequired := true) runs on NR5GC_PTC
  {
    var template (omit) GSM_MobilityInfo_Type v_GSM_MobilityInfo := f_Get_PDUSessionForDNNType(f_NR5GC_MobileInfo_GetSessionInfoList(), Emergency_PDN); // @sic R5s210363 sic@
    var GSM_MobilityInfo_Type v_PDUSession := valueof(v_GSM_MobilityInfo); // Must be there as previously established it
    var O1_Type v_PDUSessionId := v_PDUSession.SessionId;
    var boolean v_WaitForReq := true; // @sic R5s220524 sic@
    timer t_Wait := 5.0;
    
    if (p_MOTriggerRequired) {
      f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS2]); // @sic R5s220039 sic@
    }
    t_Wait.start;
    alt {         // @sic R5-216829 sic@
      [v_WaitForReq] a_NR5GC_PDUSessionReleaseReq (p_NR_CellId, v_PDUSessionId) // @sic R5s211524 sic@
         {
           t_Wait.stop;
           v_WaitForReq := false;
           repeat;
         }
      [] IMS[tsc_Index_IMS2].receive(cmr_IMS_IPCAN_Trigger)  // @sic R5s220039 sic@
         { t_Wait.stop }
      [] t_Wait.timeout   // @sic R5s220039 sic@
         { repeat }
    }

    f_NR5GC_PDUSessionRelease (p_NR_CellId,
                               v_PDUSessionId,
                               cs_GMM_GSM_Cause (omit, '00100100'B));
    f_IP_RtpRtcpLoopback(IP, v_PDUSession.PdnIndex, tsc_IMS_MediaPort_M1, omit);  // @sic R5s210363 sic@
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Emergency Call Release");
  }
  
  function f_IMS_CC_Postamble(template (omit) IMS_TestProcedure_Type p_TestProcedure := omit,
                              template (omit) IMS_TCP_CloseHandling_Type p_TcpCloseHandling := omit,
                              float p_DelayForDL_SIP := 2.0) runs on IMS_PTC
  { /* @sic R5s140801: new parameter p_TcpCloseHandling, R5s170028: removed parameter p_WaitForDeRegistration sic@ */
    /* @sic R5s190785: new parameter p_DelayForDL_SIP sic@ */
    var boolean v_AllowRetransmission := fl_IMS_PTC_CheckRanType(UTRAN_FDD) or fl_IMS_PTC_CheckRanType(UTRAN_TDD);
    var float v_DelayForDL_SIP := p_DelayForDL_SIP;

    if (isvalue(p_TestProcedure)) {
      if (valueof(p_TestProcedure) == IPCAN_EmergencyCall_NormalService or                  
            valueof(p_TestProcedure) == IPCAN_EmergencyCall_NoRegistration and f_IMS_PTC_RanTypeIsNR5GC()) { 
           // the IMS messages have already been sent, so tell IPCAN not to wait for it
           IPCAN.send(cms_IMS_IPCAN_Trigger);  
        }
      f_IMS_CC_ReleaseConnection(valueof(p_TestProcedure), p_TcpCloseHandling, v_DelayForDL_SIP); /* @sic R5s140801: p_TcpCloseHandling, R5s190785 ch. 2.2 sic@ */
      IPCAN.send(cms_IMS_IPCAN_ReleaseReq);
      if (valueof(p_TestProcedure) != IPCAN_Deregistration) { // @sic R5s211495 sic@
        f_IMS_IpcanReleaseWithOptionalDeregistration(v_AllowRetransmission);  /* @sic R5s140346 - MCC160 implementation: new function f_IMS_CC_IpcanReleaseWithOptionalDeregistration, R5s170541 sic@ */
      }
    } else {  // @sic R5s211495 sic@
      IPCAN.send(cms_IMS_IPCAN_ReleaseReq);
      f_IMS_IpcanReleaseWithOptionalDeregistration(v_AllowRetransmission);  /* @sic R5s140346 - MCC160 implementation: new function f_IMS_CC_IpcanReleaseWithOptionalDeregistration, R5s170541 sic@ */
    }
  }
  
  function f_NR5GC_IPCAN_ProcedureEnd(IMS_IPCAN_CO_ORD_PORT p_Port,
                                       NR_CellId_Type p_NR_CellId,
                                       IMS_TestProcedure_Type p_ProcedureType,
                                       TriggerResult_Type p_CallEstablishmentResult) runs on NR5GC_PTC
  {
    select (p_ProcedureType) {
      case (IPCAN_InitialRegistration) {
        // just release RRC connection
      }
      case (IPCAN_MO_IMS_Signalling, IPCAN_MT_IMS_Signalling) {
        // just release RRC connection
      }
      case (IPCAN_EmergencyCall_NormalService, IPCAN_EmergencyCall_NoRegistration) {
        f_NR5GC_EmergencyCallRelease(p_NR_CellId , false);  // If MO, UT will already have been sent);
      }
      case (IPCAN_EmergencyCall_LimitedService) {
      }
      case (IPCAN_MO_SpeechCall, IPCAN_MT_SpeechCall) {
        if (p_CallEstablishmentResult == NORMAL) {       // NOTE: in test cases for unsuccessful behaviour no speech flow is established */
          f_NR5GC_ModifyPDUSession_RemoveIMSVoice(p_NR_CellId);  // @sic R5s210096: f_NR5GC_ModifyPDUSession_RemoveIMSVoice instead of f_NR5GC_CallRelease sic@
        }
      }
      case (IPCAN_MO_VideoCall, IPCAN_MT_VideoCall) {
        if (p_CallEstablishmentResult == NORMAL) {       // @sic R5s210877 sic@
          f_NR5GC_ModifyPDUSession_RemoveIMSVideo(p_NR_CellId);
        }
      }
      case (IPCAN_XCAP_Signalling) {
        // just release RRC connection
      }
    }

  }

  function f_NR5GC_IPCAN_StartProcedure(IMS_IPCAN_CO_ORD_PORT p_Port,
                                        NR_CellId_Type p_NR_CellId,
                                        IMS_TestProcedure_Type p_ProcedureType,
                                        boolean p_ConnectionEstablished) runs on NR5GC_PTC return TriggerResult_Type
  {
    var TriggerResult_Type v_Result := NORMAL;

    select (p_ProcedureType) {
      case (IPCAN_InitialRegistration) {
        f_NR5GC_IPCAN_InitialRegistration(p_NR_CellId);
      }
      case (IPCAN_EmergencyCall_NormalService,                                  // emergency call, normal service, uses 112 as default emg number
            IPCAN_EmergencyCall_NoRegistration) {
        f_NR5GC_IPCAN_EmergencyCall(p_Port, p_NR_CellId, p_ConnectionEstablished);
      }
      case (IPCAN_EmergencyCall_LimitedService) {                               // emergency call, limited service
        /* nothing to be done */
      }
...

  function f_TC_10_1and2_NR5GC_IMS2(boolean p_GeolocationUsage := false) runs on IMS_PTC
  { /* IMS implementation for PDN2 - Emergency call */
...
    //@siclog "Step 11" siclog@
    //Step 1 of annex A.8 (MT Release of Voice Call)
    //@siclog "Step 12" siclog@
    //Step 2 of annex A.8 (MT Release of Voice Call)
    //Check: Does the UE send 200 OK for the BYE request and ends the call?
    v_ContactUrlUE := f_IMS_PTC_ImsInfo_GetContactUrl(dialog);
    f_IMS_CallReleaseMT(v_ContactUrlUE);
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 12");
    
    f_IMS_TestBody_Set(false);
    
    f_IMS_CC_Postamble(IPCAN_EmergencyCall_NormalService);
    f_IMS_IMS_SendCoOrdMsg(OtherPDN);
  }

  function f_TC_10_2_NR5GC_IMS1() runs on IMS_PTC
  { // Emergency Call with emergency registration / Success / Location information available / 5GS

    var boolean v_GeolocationUsage := pc_IMS_Geolocation;
    
    f_IMS_CC_Preamble(IPCAN_EmergencyCall, IMS_REGISTERED);
    
    // Trigger SS to execute location related functions to allow UE to get its position
    if  (v_GeolocationUsage) {
      f_IMS_IPCAN_UpdateUELocationInformation(IPCAN);
    }
      
    f_IMS_IMS_SendCoOrdMsg(OtherPDN);
    
    f_IMS_IMS_WaitForTrigger(OtherPDN); // wait for trigger after IMS call release
    
    f_IMS_CC_Postamble(IPCAN_EmergencyCall_NormalService);
  }


1. [bookmark: _Toc122434493][bookmark: _Toc295288970][bookmark: _Toc325725665][bookmark: _Toc100053411]Branches executed 
On Anritsu, the test case was executed on NR band n41 using NR ciphering nea1 and integrity algorithm nia1.
3 [bookmark: _Toc122434494][bookmark: _Toc295288971][bookmark: _Toc325725666][bookmark: _Toc100053412]Execution Log Files 
3.1 [bookmark: _Toc97137458][bookmark: _Toc100053413]Qualcomm SM 8450+SDX65 5G UE
[bookmark: _Hlk90552893]The Qualcomm SDX UE passed this test case on Anritsu Protocol Conformance Test System ME7834NR. The documentation below is enclosed as evidence of the successful test case run [1]:
· Test case execution log file:
Anritsu\ TC_10_2_LOG.xml
· PICS/PIXIT parameter file:
Anritsu\ TC_10_2_PIXIT.txt


4 [bookmark: _Toc122434496][bookmark: _Toc295288973][bookmark: _Toc325725668][bookmark: _Toc100053414]References 
	[1]
	R5s220532 - Supporting information for addition of IMS 5GS emergency call test case 10.2





